```
> ## Assignment #7 Mast 334/Math 354 Solutions:
## Problem 12 page 153
f:=x->exp(-x);
x0:=0;f0:=f(x0);
x1:=0.25;f1:=f(x1);
x2:=0.75;f2:=f(x2);
x3:=1.0;f3:=f(x3);
```

$$f := x \to \mathbf{e}^{(-x)}$$

$$x0 := 0$$

$$f0 := 1$$

$$x1 := 0.25$$

$$f1 := 0.7788007831$$

$$x2 := 0.75$$

$$f2 := 0.4723665527$$

$$x3 := 1.0$$

$$f3 := 0.3678794412$$

```
> # The equations:There are 3 intervals and 3 pieces of
spline:S0,S1,S2
S0:=x->a0+b0*(x-x0)+c0*(x-x0)^2+d0*(x-x0)^3;
S1:=x->a1+b1*(x-x1)+c1*(x-x1)^2+d1*(x-x1)^3;
S2:=x->a2+b2*(x-x2)+c2*(x-x2)^2+d2*(x-x2)^3; ## We have 12
unknowns

## Values at x_i:
EQ1:=S0(x0)=f0;
EQ2:=S0(x1)=f1;
EQ3:=S1(x1)=f1;
EQ4:=S1(x2)=f2;
EQ5:=S2(x2)=f2;
EQ6:=S2(x3)=f3;
## Derivatives must agree:

dS0:=D(S0);dS1:=D(S1);dS2:=D(S2);
EQ7:=dS0(x1)=dS1(x1);
EQ8:=dS1(x2)=dS2(x2);

## Second derivatives must agree:
d2S0:=D(dS0);d2S1:=D(dS1);d2S2:=D(dS2);
EQ9:=d2S0(x1)=d2S1(x1);
EQ10:=d2S1(x2)=d2S2(x2);

## Conditiond for "free" spline:
```

```
EQ11:=d2S0(x0)=0;
EQ12:=d2S2(x3)=0;## We have 12 equations
```

$$S0 := x \to a0 + b0\,(x - x0) + c0\,(x - x0)^2 + d0\,(x - x0)^3$$

$$S1 := x \to a1 + b1\,(x - x1) + c1\,(x - x1)^2 + d1\,(x - x1)^3$$

$$S2 := x \to a2 + b2\,(x - x2) + c2\,(x - x2)^2 + d2\,(x - x2)^3$$

$$EQ1 := a0 = 1$$

$$EQ2 := a0 + 0.25\,b0 + 0.0625\,c0 + 0.015625\,d0 = 0.7788007831$$

$$EQ3 := a1 = 0.7788007831$$

$$EQ4 := a1 + 0.50\,b1 + 0.2500\,c1 + 0.125000\,d1 = 0.4723665527$$

$$EQ5 := a2 = 0.4723665527$$

$$EQ6 := a2 + 0.25\ b2 + 0.0625\ c2 + 0.015625\ d2 = 0.3678794412$$

$$dS0 := x \to b0 + 2\ c0\,(x - x0) + 3\ d0\,(x - x0)^2$$

$$dS1 := x \to b1 + 2\ c1\,(x - x1) + 3\ d1\,(x - x1)^2$$

$$dS2 := x \to b2 + 2\ c2\,(x - x2) + 3\ d2\,(x - x2)^2$$

$$EQ7 := b0 + 0.50\ c0 + 0.1875\ d0 = b1$$

$$EQ8 := b1 + 1.00\,c1 + 0.7500\,d1 = b2$$

$$d2S0 := x \to 2\ c0 + 6\ d0\,(x - x0)$$

$$d2S1 := x \to 2\ c1 + 6\ d1\,(x - x1)$$

$$d2S2 := x \to 2\ c2 + 6\ d2\,(x - x2)$$

$$EQ9 := 2\ c0 + 1.50\ d0 = 2\ c1$$

$$EQ10 := 2\ c1 + 3.00\ d1 = 2\ c2$$

$$EQ11 := 2\ c0 = 0$$

$$EQ12 := 2\ c2 + 1.50\ d2 = 0$$

```
> ## Now we solve for a0,b0,c0,d0,a1,b1,c1,d1,a2,b2,c2,d2:
solution:=solve({EQ1,EQ2,EQ3,EQ4,EQ5,EQ6,EQ7,EQ8,EQ9,EQ10,E
Q11,EQ12},
        {a0,b0,c0,d0,a1,b1,c1,d1,a2,b2,c2,d2});
```

$$solution := \{\ b1 = -0.8071887169\ ,\ a1 = 0.7788007831\ ,\ a2 = 0.4723665527\ ,\ a0 = 1.,$$
$$c0 = 0.,\ b0 = -0.9236009430\ ,\ b2 = -0.4570524007\ ,\ d1 = -0.1540167840\ ,$$
$$c2 = 0.2346237282\ ,\ c1 = 0.4656489042\ ,\ d2 = -0.3128316376\ ,\ d0 = 0.6208652056\ \}$$

```
> assign(solution);## This command assigns the values found to the coefficients a0,...,d2.
> S:=x->piecewise(x<0.25,S0(x),x<0.75,S1(x),S2(x));## We glue the pieces together
```

$$S := x \to \text{piecewise}(x < 0.25, S0(x), x < 0.75, S1(x), S2(x))$$

```
> plot([f,S],0..1);## Visual comparison of f and the approximating spline
```




```
> int01:=int(S(x),x=0..1);##integral of the spline
err:=evalf(abs(int01-1+1/exp(1)));## the error
```

$$int01 := 0.6319663611$$

$$err := 0.0001541977$$

```
> ## Since 0.5 is between 0.25 and 0.75 we have to use S1:
df:=D(f);d2f:=D(df);## derivatives of the function f
dS1(0.5);err1:=evalf(abs(dS1(0.5)-df(0.5)));
d2S1(0.5);err2:=evalf(abs(d2S1(0.5)-d2f(0.5)));
```

$$df := x \to -\mathbf{e}^{(-x)}$$

$$d2f := x \to \mathbf{e}^{(-x)}$$

-0.6032424119

$$err1 := 0.0032882478$$

0.7002726324

$$err2 := 0.0937419727$$

```
>
```